

## a HA függvény - különlegességek

### FOGALMAK

A HA függvény az első argumentumával meghatározott feltétel teljesülését vizsgálja. Ha a feltétel teljesül, akkor végrehajtja a második argumentumával deklarált műveletet. Ha nem, akkor a harmadik argumentumával meghatározottat. Az első két argumentum kötelező. Ha a feltétel nem teljesül és a felhasználó nem deklarált „hamis műveletet”, akkor a függvény HAMIS logikai értéket ad eredményül. Ez a függvény szokásos használata. A nem szokásos... Erről szól a cikk.

### EGYETLEN FELTÉTEL

De először vegyük az alapesetet. A függvény első argumentuma tehát egy feltétel, de fogalmazhatnánk úgy is, hogy egy tetszőleges kifejezés, amely logikai értéket ad eredményül: IGAZ-at vagy HAMIS-at.

	A	B	C
1	eredeti szám	módosított szám	
2	70	35	
3	40	80	
4	30	60	
5	10	20	
6	60	30	
7	50	25	
8			
9	B2 » =HA( A2<50 ; A2*2 ; A2/2 )		
10			
11			

	A	B	C	D
1	munkadíj	kifizette	bevétel	
2	515 000	2020.05.08	515 000	
3	420 000		tartozás	
4	420 000	2020.08.13	420 000	
5	655 000	2020.09.27	655 000	
6	330 000		tartozás	
7	690 000	2020.02.02	690 000	
8				
9				
10	C2 » =HA( ÜRES(B2) ; "tartozás" ; A2 )			
11				
12				

A HA függvény első argumentumában álló dátum- és idő konstansokat a DÁTUMÉRTÉK és az IDŐÉRTÉK függvényekkel számmá kell alakítanunk. Ez a szabály a logikai függvényekre is vonatkozik! A két függvény egyetlen argumentuma a szöveggént megadott dátum illetve idő.

	A	B	C	D	E	F
1	dátum	késés (nap)				
2	2020-10-25					
3	2020-10-24					
4	2020-10-16					
5	2020-10-10	6				
6	2020-10-05	11				
7	2020-10-26					
8						
9	B2 » =HA( A2<DÁTUMÉRTÉK("20/10/16") ; "20/10/16"-A2 ; "" )					
10						
11						

	A	B	C	D	E	F
1	érkezés	késés (p.)				
2	9:19:00	19				
3	8:37:00	0				
4	9:02:00	2				
5	9:14:00	14				
6	8:51:00	0				
7	8:34:00	0				
8						
9	B2 » =HA( A2>IDŐÉRTÉK("9:00:00") ; (A2-"9:00:00")*24*60 ; 0 )					
10						
11						

Az Excelben kevésbé járatos felhasználók körében gyakran előfordul a hivatkozott logikai érték relációba helyezése. Megmutatom miről van szó!

	A	B	C	D	E
1	kolléga	fizetés	dicséret	prémium	
2	Sas Pál	265 000	IGAZ	132 500	
3	Réz Ede	295 000	HAMIS		
4	Ács Tas	380 000	IGAZ	190 000	
5	Sós Örs	445 000	HAMIS		
6	Pék Ida	320 000	HAMIS		
7	Gál Éva	475 000	IGAZ	237 500	
8					
9		D2 » =HA( C2 ; B2*50% ; "" )			
10					
11					

A dicséretben részesült kollégák prémiumot kapnak, amely a fizetésük fele. A HA függvény első argumentuma tehát egy hivatkozás, amellyel arra utasítjuk a programot, hogy olvassa ki a logikai értéket a cellából. Tehát, nem azt mondjuk, vizsgáld meg a cellát, IGAZ-t tartalmaz-e, mert a vizsgálat eredménye azonos lenne a cella tartalmával.

C2	C2 = IGAZ	eredmény
IGAZ	IGAZ = IGAZ	IGAZ
HAMIS	HAMIS = IGAZ	HAMIS

A függvény akkor is működik, ha az első argumentumának kiértékelése nem logikai értéket, hanem egyet vagy nullát ad eredményül. A nulla a HAMIS-sal, az egy az IGAZ-zal egyenértékű.

	A	B	C	D	E
1	szám	módosítva			
2	30	31			
3	28	29			
4	26	27			
5	21	20			
6	20	19			
7	45	44			
8					
9	B2 » =HA( KEREKÍTÉS( VÉL(); 0 ) ; A2 + 1 ; A2-1 )				
10					
11					

Sőt! Elárulok valamit. De, maradjon titok! Nem csak eggyel és nullával, de bármilyen számmal működik a függvény: HA( <szám> ; <művelet, ha szám nem nulla> ; <művelet, ha szám nulla> ).

Figyelem, figyelem! A HA( <logikai kifejezés> ; IGAZ ; HAMIS ) formula alkalmazása amatőrség! Az ilyen típusú vizsgálatokhoz ugyanis nincs szükség a HA függvényre. A program, még az összetett logikai kifejezésekkel is gond nélkül elboldogul és IGAZ-at vagy HAMIS-at ad eredményül.

#### FELTÉTEL-RENDSZER

Az ÜRES függvényes példa, elgondolkodtató... Logikai értéket eredményező függvényt is lehet használni?! Hát, akkor több feltételt is deklarálhatunk! Vegyük számba a szóba jöhető függvényeket!

	A	B	C	D	E	F	G	H
1	alak	vékony	darab		VAGY	ÉS	XVAGY	
2	ellipszis	IGAZ	7		IGAZ	HAMIS	HAMIS	
3	ellipszis	HAMIS	6		IGAZ	HAMIS	IGAZ	
4	négyzet	HAMIS	3		HAMIS	HAMIS	HAMIS	
5	kör	IGAZ	7		IGAZ	IGAZ	IGAZ	
6	ellipszis	HAMIS	4		HAMIS	HAMIS	HAMIS	
7	kör	HAMIS	9		IGAZ	HAMIS	HAMIS	
8	kör	HAMIS	8		IGAZ	HAMIS	HAMIS	
9	ellipszis	IGAZ	1		IGAZ	HAMIS	IGAZ	
10	négyzet	IGAZ	1		IGAZ	HAMIS	IGAZ	
11								
12		E2-F2-G2 » =VAGY-ÉS-XVAGY( A2="kör" ; B2 ; C2>5 )						
13								

A bal oldali táblázat sorait vizsgáltattam meg a többargumentumos logikai függvényekkel. Három feltételt határoztam meg. Legyen kör, legyen vékony és a daraszám legyen legalább hat. A kritériumoknak megfelelő cellákat szürkével színeztem. A jobb oldali tábla a vizsgálat eredményét tartalmazza. Az IGAZ eredményeket kékkel jelöltem.

Tehát a feltétel-rendszer kialakításához három függvényt vehetünk igénybe. A beírás módja szokásos: =HA( VAGY/ÉS/XVAGY( <kifejezés<sub>1</sub>> ; <kifejezés<sub>2</sub>> ; <kifejezés<sub>3</sub>> ... <kifejezés<sub>255</sub>> ) ; <igaz-művelet> ; <hamis-művelet> ).


A három függvény argumentumai logikai eredményt adó kifejezések. A VAGY akkor ad IGAZ-at, ha legalább egy argumentuma IGAZ. Az ÉS csak akkor, ha minden argumentuma IGAZ. Az XVAGY pedig akkor, ha argumentumainak göngyöltett kiértékelése végül ellentétes logikai értékeket eredményez. Elmagyarázom!

„Versikéje” alapján az XVAGY-ot kizáró VAGY-nak is hívhatjuk: vagy az egyik, vagy a másik. Más variáció nincs. Igen ám, de az XVAGY-nak kettőnél több argumentuma is lehet! Ezt az ellentmondást a program göngyöltett kiértékeléssel oldja fel: az argumentumok kiértékelése balról jobbra halad, mindig csak két tényezőt figyelembe véve. Az első az eddigi lépések göngyöltett eredménye a másik a következő argumentum kiértékelésének eredménye. Valahogy így:

=XVAGY( 3 > 2 ; 2 = 2 ; 1 <> 1 ; 2 <= 1 ; 2 >= 3 )				
3 > 2	2 = 2	1 <> 1	2 <= 1	2 >= 3
IGAZ	IGAZ			
HAMIS		HAMIS		
	HAMIS		IGAZ	
		IGAZ		HAMIS
			IGAZ	

Amióta az „ismeretlen programozó”, aki nem más, mint jakutföldi Fegyva, észrevette, hogy ha az IGAZ értéket adó argumentumok száma páratlan, akkor a végeredmény mindig IGAZ és ha páros, akkor meg mindig HAMIS, szóval amióta észrevette, azóta csak ezt lehet olvasni minden ismertetőben.

Feltétel-rendszer segítségével tudjuk elrejteni az előre elkészített, de még feldolgozandó adatok nélküli képlet eredményét is. Miről van szó! Előkészítünk egy táblázatot, meghatározott tevékenység támogatására. Például, egy kisvállalat házi-pénztárának adminisztrálására. Elkészítjük a feliratókat, létrehozuk az adatbázis-táblázatot (*Beszűrés, Táblázatok, Táblázat*), megadjuk a beviteli listák elemeit (*Adatok, Adateszközök, Érvényesítés*) és megadjuk a cella- és szám-formátumokat.



A „felvétel” az, amikor elviszi a pénzt a kolléga, és az „elszámolás”, amikor visszahozza a maradékot és a számlát. Az egyenleg oszlop képletében tehát két műveletet kell deklarálnunk. Ha az akció felvétel, akkor az előző egyenlegből, le kell vonnunk a felvett összeget. Ha elszámolás, akkor az előző egyenleghez hozzá kell adnunk a visszahozott összeget.

Az első akció képlete azonban ettől eltérő. Ha a tulajdonos, mondjuk három-milliót tett a kasszába, akkor az első egyenleg három-millió mínusz a felvett összeg. Mivel az adatbázis-táblázat oszlopában csak egyetlen képletet határozhatunk meg, ezért az első akció egyenlegének kiszámítását is bele kell foglalnunk a képletbe! A művelet feltétele legyen a képletet tartalmazó sor száma: ha kető, akkor legyen ez, különben meg az.

Kérjünk még egy sort és az új sor egyenleg cellájában dolgozzunk. Azért ott, hogy ne a táblázat fejlécére kelljen kattintanunk, amikor az előző egyenlegre hivatkozunk.

	A	B	C	D	E	F
1	név	dátum	akció	összeg	egyenleg	
2					3000000	
3					3000000	
4						
5						

```
=HA( SOR()=2 ; 3000000-[@összeg] ; HA([@akció]="felvétel" ; E2-[@összeg] ; E2+[@összeg] ) )
```

A program, természetesen az első sorba is elhelyezi a képletet, ezért a szerkesztéshez használt sort a művelet követően letörölhetjük, és a képletet kiegészíthetjük a rejtéssel.


=HA(ÉS([@név]<>"";[@ dátum]<>"";[@akció]<>"";[@összeg]<>""];<egyenleg>;""))

Az <egyenleg> az előbb ismertett képletet jelöli. Összefoglalva: az aktuális egyenleg csak akkor fog megjelenni, ha az ÉS függvény argumentumaiban szereplő cellák ki lesznek töltve.

#### HA FÜGGVÉNYEK EGYMÁSBA ÁGYAZÁSA

Ha azt mondanám, a HA függvény műveletek közötti választás automatizálására szolgál, akkor olyan nagyon nem lőnék mellé. Alapesetben két művelet között „választ”, egymásba ágyazottan legfeljebb hatvanöt közül.

=HA(A1=1;"egy";HA(A1=2;"kettő";HA(A1=3;"három";HA(A1=4;"négy";HA(A1=5;"öt";HA(A1=6;"hat";HA(A1=7;"hét";HA(A1=8;"nyolc";HA(A1=9;"kilenc";HA(A1=10;"tíz";HA(A1=11;"tizenegy";HA(A1=12;"tizenkettő";HA(A1=13;"tizenhárom";HA(A1=14;"tizennégy";HA(A1=15;"tizenöt";HA(A1=16;"tizenhat";HA(A1=17;"tizenhét";HA(A1=18;"tizennyolc";HA(A1=19;"tizenkilenc";HA(A1=20;"húsz";HA(A1=21;"huszonegy";HA(A1=22;"huszonkettő";HA(A1=23;"huszonhárom";HA(A1=24;"huszonnégy";HA(A1=25;"huszonöt";HA(A1=26;"huszonhat";HA(A1=27;"huszonhét";HA(A1=28;"huszonnolc";HA(A1=29;"huszonkilenc";HA(A1=30;"harminc";HA(A1=31;"harmincegy";HA(A1=32;"harminckettő";HA(A1=33;"harminchárom";HA(A1=34;"harmincnégy";HA(A1=35;"harmincöt";HA(A1=36;"harminchat";HA(A1=37;"harminchét";HA(A1=38;"harmincnyolc";HA(A1=39;"harminckilenc";HA(A1=40;"negyven";HA(A1=41;"negyvenegy";HA(A1=42;"negyvenkettő";HA(A1=43;"negyvenhárom";HA(A1=44;"negyvennégy";HA(A1=45;"negyvenöt";HA(A1=46;"negyvenhat";HA(A1=47;"negyvenhét";HA(A1=48;"negyvennyolc";HA(A1=49;"negyvenkilenc";HA(A1=50;"ötven";HA(A1=51;"ötvenegy";HA(A1=52;"ötvenkettő";HA(A1=53;"ötvenhárom";HA(A1=54;"ötvennégy";HA(A1=55;"ötvenöt";HA(A1=56;"ötvenhat";HA(A1=57;"ötvenhét";HA(A1=58;"ötvennyolc";HA(A1=59;"ötvenkilenc";HA(A1=60;"hatvan";HA(A1=61;"hatvanegy";HA(A1=62;"hatvankettő";HA(A1=63;"hatvanhárom";HA(A1=64;"hatvannégy";"hatvanöt")))



Lefordítom ómagyarra! A beírt képletet a program nem tudja kiértékelni, mert az 64 beágyazási szintnél többet tartalmaz.

Nézzünk egy beágyazós feladatot, amelynek képletében három HA függvénnyel négy műveletet kezelünk! A hallgatók dolgozatot írtak. Az értékelés a pontszámok alapján történik: 0-tól 39-ig „értékelhetetlen”, 40-től 79-ig „kevés”, 80-tól 179-ig „megfelelt”, 180-tól „dicséretes”.

	A	B	C	D
1	hallgató	pontszám	értékelés	
2	Alföldi Vera	67	kevés	
3	Erdélyi Máté	171	megfelelt	
4	Nyitrai Géza	51	kevés	
5	Nógrádi Hugó	21	értékelhetetlen	
6	Pataki Dóra	123	megfelelt	
7	Mohácsi Rita	196	dicséretes	
8				
9				

=HA( B2 < 40 ; "értékelhetetlen" ; HA( B2 < 80 ; "kevés" ; HA( B2 < 180 ; "megfelelt" ; "dicséretes" )))

Általánosságban is megfogalmazhatjuk a HA függvény egymásba ágyazott képletét: HA( <feltétel1> ; <művelet1> ; HA( <feltétel2> ; <művelet2> ; HA( <feltétel3> ; <művelet3> ... HA( <feltételn> ; <műveletn> ; <műveletn+1> ) ) ... )

## A HA FÜGGVÉNY TÖMBÖS KIÉRTÉKELÉSE

A HA függvény tehát kiértékel egy logikai kifejezést és ha IGAZ eredményre jut, akkor az egyik műveletet hajtja végre, ha HAMIS-ra, akkor a másikat. Ha ezt a feltételes művelet-végzést „haművelet”-nek nevezzük, akkor azt mondhatjuk, hogy a tömbös kiértékelésű HA függvénnyel több ha-művelet elvégzésére utasítjuk a programot. A ha-műveletek eredményét a függvény egy tömbben adja vissza. Aki nem tudja mi az a tömb és mi az a tömbös kiértékelés, az [tájékozódjon!](#)

	A	B	C	D	E
1	név	születési hely	lakhely	életkor	
2	Polyák Amália	Felsőzsolca	Felsőzsolca	60	
3	Lugosi Adalbert	Tiszakécske	Pétervására	46	
4	Barta Ivó	Salgótarján	Kiskunhalas	43	
5	Rozsnyai Bátor	Simontornya	Tiszakécske	59	
6	Somos Júlia	Soltvadkert	Soltvadkert	31	
7	Szántai Marietta	Nagykanizsa	Kiskunmajsa	42	
8	Gazsó Báborka	Lengyeltóti	Lengyeltóti	59	
9					
10					

{=MAX( HA( B2:B8 = C2:C8; D2:D8))} » 60  
 {60; HAMIS; HAMIS; HAMIS; 31; HAMIS; 59}

A képen látható példa kérdése az volt: Hány éves a hajóskapitány? Akarom mondani, a legöregebb ember, aki a születési helyén lakik? A HA függvénnyel ezt az utasítást adtuk a programnak: ellenőrizd soronként, azonos-e a születési hely és a lakhely, ha igen jegyezd meg ennek a sornak az életkorát! A képen alul láthatjuk a HA függvény eredmény-tömbjét. A feltételnek nem megfelelő sorok HAMIS értékeit a második művelet hiánya magyarázza.

Persze kérdezhették volna azt is, hogy ki az a legöregebb ember, aki a születési helyén lakik. És nem az emberek életkorát, hanem a születési dátumát adják meg. Akkor így alakulna a megoldás.

	A	B	C	D	E
1	név	születési hely	lakhely	születés	
2	Polyák Amália	Felsőzsolca	Felsőzsolca	1960.07.14	
3	Lugosi Adalbert	Tiszakécske	Pétervására	1974.05.03	
4	Barta Ivó	Salgótarján	Kiskunhalas	1977.05.19	
5	Rozsnyai Bátor	Simontornya	Tiszakécske	1961.01.06	
6	Somos Júlia	Soltvadkert	Soltvadkert	1989.02.25	
7	Szántai Marietta	Nagykanizsa	Kiskunmajsa	1978.11.10	
8	Gazsó Báborka	Lengyeltóti	Lengyeltóti	1961.07.21	
9					
10					

ha-műveletek eredménytömbje  
 {=HA( B2:B8 = C2:C8; MA()-D2:D8 ; 0 )} » { 22014 ; 0 ; 0 ; 0 ; 11561 ; 0 ; 21642 }

az eredménytömb legnagyobb száma  
 {=MAX( HA( B2:B8 = C2:C8; MA()-D2:D8 ; 0 ))} » 22014

a legnagyobb szám indexe a tömbben  
 {=HOL.VAN( MAX( HA( B2:B8 = C2:C8; MA()-D2:D8 ; 0 )) ;  
 HA( B2:B8 = C2:C8; MA()-D2:D8 ; 0 ))} » 1

az index-szel azonos sorszámú név kiírása  
 {=INDEX( A2:A8 ; HOL.VAN( MAX( HA( B2:B8 = C2:C8; MA()-D2:D8 ; 0 )) ;  
 HA( B2:B8 = C2:C8; MA()-D2:D8 ; 0 )))} » Polyák Amália

A legidősebb embert most a megélt napok számával határoztuk meg. Ebben a példában már deklaráltunk hamis-műveletet: ha nem azonos a születési- és a lakhely, akkor az eredmény legyen nulla. A tömbös feldolgozás most is egyetlen adatot eredményezett, de ennek ellenére nem mellőzhetjük

a képletet lezáró *Shift+Ctrl+Enter* billentyű-parancsot! A #HIÁNYZIK hibaérték valószínűleg erre figyelmeztet.

### FELTÉTEL-RENDSZER TÖMBÖS KIÉRTÉKELÉSŰ HA FÜGGVÉNYBEN

A logikai függvényeket, beleértve a NEM-et is, nem használhatjuk tömbös kiértékelésű HA függvényben. Ennek ellenére nem kell lemondanunk a függvény összetett feltételezéséről. Csak egy speciális írásmódot kell használnunk.

A logikai kifejezéseket külön-külön zárójeleznünk kell. A kifejezések között logikai AND vagy logikai OR kapcsolatot hozhatunk létre. Az előbbinek a szorzás (\*), az utóbbinak az összeadás (+) a műveleti jele. A kifejezések kiértékelése balról jobbra halad. Természetesen „plusz” zárójelezéssel a végrehajtási sorrend módosítható. Nézzünk néhány példát az egyszerűből az összetett felé haladva.

	A	B	
1	alak	vékony	
2	kör	IGAZ	
3	négyszög	IGAZ	
4	ellipszis	HAMIS	
5	négyszög	IGAZ	
6	ellipszis	IGAZ	
7	négyszög	HAMIS	
8	ellipszis	HAMIS	
9	rombusz	HAMIS	
10	kör	IGAZ	
11	ellipszis	HAMIS	
12			

	A	B	
1	alak	vékony	
2	kör	IGAZ	
3	négyszög	IGAZ	
4	ellipszis	HAMIS	
5	négyszög	IGAZ	
6	ellipszis	IGAZ	
7	négyszög	HAMIS	
8	ellipszis	HAMIS	
9	rombusz	HAMIS	
10	kör	IGAZ	
11	ellipszis	HAMIS	
12			

	A	B	C	
1	sorszám	alak	vékony	
2	01	kör	IGAZ	
3	02	négyszög	IGAZ	
4	3	ellipszis	HAMIS	
5	04	négyszög	IGAZ	
6	05	ellipszis	IGAZ	
7	6	négyszög	HAMIS	
8	07	ellipszis	HAMIS	
9	08	rombusz	HAMIS	
10	9	kör	IGAZ	
11	10	ellipszis	HAMIS	
12				

a vékony nem ellipszisek száma $\{=SZUM(HA((A2:A11<>"ellipszis")*(B2:B11));1;0)\} \gg 4$
a nem vékony körök és ellipszisek száma $\{=SZUM(HA(((A2:A11="kör")+(A2:AB11="ellipszis"))*(1-B2:B11));1;0)\} ; 0)\} \gg 3$
a vékony körök és ellipszisek száma, minden harmadik sort vizsgálva $\{=SZUM(HA((MARADÉK(A2:A11;3)=0)*((B2:B11="kör")+(B2:B11="ellipszis"))*(C2:C11));1;0)\} \gg 1$

A feladatunk mindhárom példában a feltétel-rendszernek eleget tévő sorok megszámlálása volt. Ezért deklaráltunk egyet (1) igaz-műveletként és nullát (0) hamis-műveletként. Majd a HA függvény eredmény-tömbjében álló számokat a SZUM-mal adtuk össze.

A második példában a  $1-B2:B11$  kifejezés a  $B2:B11=HAMIS$  „szinonimája”, vagy másként fogalmazva, a  $NEM(B2:B11)$  kifejezés függvény nélküli változata. A harmadik példában szereplő MARADÉK függvény két argumentuma az osztandó és az osztó. Eredménye a két szám osztási maradéka.

Szabadság, egyenlőség, testvériség!



margitfalvi.arpad@gmail.com